

Computer-Aided Electroacoustic Design with SPICE*

W. MARSHALL LEACH, JR., *Fellow*

Georgia Institute of Technology, School of Electrical Engineering, Atlanta, GA 30332-0250, USA

The application of the computer program SPICE to the modeling of electroacoustic systems is described. The basic SPICE models for acoustic mass, resistance, and compliance; mechanical mass, resistance, and compliance; electrical-to-mechanical transducers; and mechanical-to-acoustical transducers are given. Several examples are presented of the analysis of microphones, loudspeakers, and crossover networks.

0 INTRODUCTION

SPICE is a widely used computer program, which was developed as an aid for the design of solid-state electronic circuits with an emphasis on integrated circuits. In addition to its applications to solid-state circuits, the program has the capability to analyze complex linear circuits containing independent and controlled sources, resistors, inductors, and capacitors. Because these are the basic elements of electroacoustic modeling, SPICE can be used for the analysis of electroacoustic systems. Several versions of the program are available for the personal computer, which have powerful graphics software, thus making possible a rapid evaluation of the frequency response, impedance, and transient response of electroacoustic systems such as microphones and loudspeakers.

This engineering report presents an introduction to the use of SPICE as an aid to electroacoustic system design. SPICE models are given for the basic electrical, mechanical, and acoustical elements of electroacoustic modeling and for the transducer elements that couple electrical to mechanical systems and mechanical to acoustical systems. Several examples are presented which illustrate the application of the models to the areas of microphone design, loudspeaker design, and crossover network design. One of the features of the modeling is that controlled sources are used in place of the usual transformer. This makes it possible to use either impedance or mobility models in the analogous circuits. Only the impedance analogous circuits are covered here.

The scope of this report is to provide an introduction to the application of SPICE to electroacoustic problems.

* Manuscript received 1991 March 19.

It is not intended to be a complete reference on the subject. A good description of the analysis capabilities of SPICE can be found in [1] and [2]. Although these references concentrate primarily on electronic circuit applications, they are useful in the present context also.

1 INDEPENDENT SOURCES IN SPICE

The SPICE symbols for a constant-voltage source and a constant-current source are V and I, respectively. Fig. 1 illustrates the SPICE circuit models for each, where the conventions for the positive (N+) and negative (N-) reference nodes are labeled. For an ac sinusoidal steady-state analysis, the codes that specify these sources are

$$\text{VX N+ N- AC VALUE} \quad (1)$$

$$\text{IX N+ N- AC VALUE} \quad (2)$$

where X is the name for the source, N+ and N- are the positive and negative source nodes, respectively, and VALUE is the ac numerical value of the source. The instruction for performing an ac analysis in decade intervals is

$$\text{.AC DEC ND FSTART FSTOP} \quad (3)$$




Fig. 1. (a) Constant-voltage source. (b) Constant-current source.

where ND is the number of points per decade, FSTART is the start frequency, and FSTOP is the stop frequency.

For a transient pulse analysis, the codes that specify voltage and current sources are

VX N+ N- PULSE(V1 V2 TD TR TF PW PER) (4)

IX N+ N- PULSE(I1 I2 TD TR TF PW PER) (5)

where V1 (I1) is the initial value of the pulse, V2 (I2) the pulsed value, TD the time delay, TR the rise time, TF the fall time, PW the pulse width, and PER the period. The instruction for performing a transient pulse analysis is

.TRAN TSTEP TSTOP (6)

where TSTEP is the printing increment and TSTOP is the final time. There are other types of sources modeled by SPICE besides the ac sinusoidal steady-state source and the pulse source. These are not covered here.

2 PASSIVE TWO-TERMINAL ELEMENTS IN SPICE

The passive two-terminal components that are modeled by SPICE are the resistor R , the inductor L , and the capacitor C . The respective SPICE codes that specify these elements are

RX N1 N2 VALUE (7)

LX N1 N2 VALUE (8)

CX N1 N2 VALUE (9)

where X is the name for the element, N1 and N2 are the circuit nodes to which the two leads of the element connect, and VALUE is the numerical value of the element.

3 CONTROLLED SOURCES IN SPICE

There are four controlled sources in SPICE: the voltage-controlled voltage source ($V2 = E \times V1$), the current-controlled voltage source ($V2 = H \times I1$), the current-controlled current source ($I2 = F \times I1$), and the voltage-controlled current source ($I2 = G \times V1$). The SPICE code for the voltage-controlled voltage source is of the form

EX N+ N- NC+ NC- VALUE (10)

where X is the name for the source, N+ and N- are the positive and negative terminals, respectively, NC+ and NC- are the positive and negative controlling nodes, respectively, and VALUE is the numerical value of the voltage gain E of the source. The SPICE code

for the current-controlled voltage source is of the form

HX N+ N- VNAME VALUE (11)

where X is the name for the source, N+ and N- are the positive and negative terminals, respectively, VNAME is the name of the voltage source through which the controlling current flows, and VALUE is the numerical value of the transresistance gain H of the source.

The SPICE code for the current-controlled current source is of the form

FX N+ N- VNAME VALUE (12)

where X is the name for the source, N+ and N- are the positive and negative terminals, respectively, VNAME is the name of the voltage source through which the controlling current flows, and VALUE is the numerical value of the current gain F of the source. The SPICE code for the voltage-controlled current source is of the form

GX N+ N- NC+ NC- VALUE (13)

where X is the name for the source, N+ and N- are the positive and negative terminals, respectively, NC+ and NC- are the positive and negative controlling nodes, respectively, and VALUE is the numerical value of the transconductance gain G of the source.

The conventions for the positive (N+ and NC+) and negative (N- and NC-) reference nodes in the codes for the four controlled sources follow those used in Fig. 1. The current-controlled voltage source and the current-controlled current source each require the name of a voltage source through which the controlling current flows. These sources must be constant-voltage sources. If the controlling current flows in a branch in which there is no constant-voltage source that can be specified, a source with an assigned voltage of zero can be inserted in series with the branch. Thus a constant-voltage source with an assigned value of zero can be thought of as an ammeter in SPICE.

4 SPICE MODELS FOR INDEPENDENT ELECTROACOUSTIC SOURCES

There are six independent sources that are used in the analogous circuits of electroacoustic modeling. The constant-voltage generator and the constant-current generator are the two independent sources in the electrical parts of the analogous circuits. The constant-force generator and the constant-velocity generator are the two independent sources in the mechanical parts. The constant-pressure generator and the constant-volume-velocity generator are the two independent sources in the acoustical parts. The SPICE models for the impedance analogs of these six sources are given in this section.

In the electrical parts of electroacoustic models,

voltage is denoted by e and current by i . In the mechanical parts, force is denoted by f and velocity by u . In the acoustical parts, pressure is denoted by p and volume velocity by U . Table 1 lists the SPICE generator that is used to model independent sources for each of these variables in impedance analogous circuits.

5 SPICE MODELS FOR PASSIVE TWO-TERMINAL ELECTROACOUSTIC ELEMENTS

In the electrical parts of electroacoustic models, the passive two-terminal elements are the resistor, the inductor, and the capacitor. Let these elements be denoted by R_E , L_E , and C_E , respectively. In the mechanical parts, the passive two-terminal elements are the damping resistance, the mass, and the compliance, denoted by R_M , M_M , and C_M , respectively. In the acoustical parts, the passive two-terminal elements are the acoustic resistance, the acoustic mass, and the acoustic compliance, denoted by R_A , M_A , and C_A , respectively. Table 2 lists the SPICE element that is used to model each of these elements in impedance analogous circuits.

6 SPICE MODELS FOR ELECTROACOUSTIC TRANSDUCER ELEMENTS

There are four transducer elements that are often encountered in electroacoustic modeling. These are the electromagnetic-mechanical transducer, the crystal electrostatic-mechanical transducer, the condenser electrostatic-mechanical transducer, and the mechano-acoustic transducer. Conventional models for these usually involve the use of transformers. Because of the restrictions imposed by the electrical relations between the transformer terminal variables, mobility models can be required in either the mechanical or the

acoustical parts of the system model, or in both. The gyrator is a circuit element that has been used in place of the transformer to circumvent mobility models. In this report, controlled sources are used to represent transducer elements. With the controlled sources, restrictions imposed by the transformer terminal relations can be avoided and impedance and mobility analogs can be used interchangeably. Only the impedance models are given here.

6.1 Electromagnetic-Mechanical Transducer [3]

A simplified form of this transducer is illustrated in Fig. 2(a), where a length of straight wire is shown cutting in a magnetic field B . In practice, the transducer usually consists of a coil of wire rather than a straight length. For this reason, it is often called a moving-coil transducer. It is used in the modeling of the dynamic microphone and the dynamic loudspeaker. Let the effective length of the wire that cuts the magnetic field be denoted by l . The basic electromechanical equations that relate the back EMF e in the coil to the velocity u with which it moves and the force f exerted on the coil to the current i that flows through it are

$$e = Blu \tag{14}$$

$$f = Bli \tag{15}$$

The impedance analogous controlled-source model for the electromagnetic-mechanical transducer [4] is given in Fig. 2(b). The circuit can be implemented in SPICE with a current-controlled voltage source and a current-controlled current source as described.

Table 1. Impedance analogous SPICE models for independent sources.

Electroacoustic variable	SPICE model impedance analog
e	VX
i	IX
f	VX
u	IX
p	VX
U	IX

Table 2. Impedance analogous SPICE models for passive two-terminal elements.

Electroacoustic element	SPICE element impedance analog
R_E	RX
L_E	LX
C_E	CX
R_M	RX
M_M	LX
C_M	CX
R_A	RX
M_A	LX
C_A	CX




Fig. 2. (a) Electromagnetic-mechanical transducer. (b) Controlled-source model of transducer.

6.2 Crystal Electrostatic-Mechanical Transducer [3]

This transducer is sometimes called the piezoelectric transducer. It is the basic transducer used in the modeling of the crystal microphone and the crystal loudspeaker. Fig. 3(a) illustrates the basic construction of the transducer. It consists of a piezoelectric crystal element mounted against a rigid surface with capacitive plates attached to two opposite sides. The basic electromechanical equations that relate the terminal voltage e and current i , the force f exerted by the face of the crystal, and the velocity u with which the face of the crystal moves are

$$i = C_E s e + \tau C_E u \tag{16}$$

$$u = -C_M s f + \tau C_M i \tag{17}$$

where s is the complex frequency, C_E the electrical capacitance between the plates on the crystal, C_M the mechanical compliance of the crystal, and τ the crystal coupling coefficient. The impedance analogous controlled-source circuit model for the crystal electrostatic-mechanical transducer [4] is given in Fig. 3(b). The circuit can be implemented in SPICE with two capacitors and two current-controlled current sources as described.

6.3 Condenser Electrostatic-Mechanical Transducer [3]

This is the transducer that is used in modeling condenser microphones and electrostatic loudspeakers. Two

types of condenser transducers are covered here—the single ended and the push-pull. The former is normally used in microphones, while the latter is used in loudspeakers. Fig. 4(a) illustrates the basic construction of the single-ended transducer as it might appear in a microphone. It consists of a thin conducting diaphragm suspended in front of a conducting back plate. Insulating material is used between the diaphragm and the back plate so that a dc polarizing voltage can be applied between the two.

The basic electromechanical equations that relate the small-signal components of the terminal voltage e and current i , the small-signal force f exerted by the diaphragm, and the small-signal velocity u with which the diaphragm moves are

$$i = C_{E0} s e - \frac{E C_{E0}}{x_0} u \tag{18}$$

$$u = -C_M s f - \frac{E C_M}{x_0} i \tag{19}$$

where s is the complex frequency, E the dc polarizing voltage between diaphragm and back plate, C_{E0} the zero-signal electrical capacitance between diaphragm and back plate, C_M the mechanical compliance of the diaphragm, and x_0 the zero-signal spacing between diaphragm and back plate. The impedance analogous controlled-source circuit model for the condenser electrostatic-mechanical transducer [4] is given in Fig. 4(b). The circuit can be implemented in SPICE with




Fig. 3. (a) Crystal electrostatic-mechanical transducer. (b) Controlled-source model of transducer.




Fig. 4. (a) Condenser electrostatic-mechanical transducer. (b) Controlled-source model of transducer.

two capacitors and two current-controlled current sources as described.

Fig. 5(a) illustrates the construction of the push-pull condenser transducer as it might appear in an electrostatic loudspeaker. The diaphragm is the center element. The two end plates are perforated to allow dipole radiation from the diaphragm. Two polarizing voltages are applied to the transducer, a positive voltage to one end plate and a negative voltage to the other. Opposite-polarity voltages $+(E + e_1)$ and $-(E + e_2)$ are applied to the perforated front and back plates, respectively, where E is the dc polarizing voltage and e_1 and e_2 are signal voltages. In normal applications, e_1 and e_2 are differential voltages, that is, $e_2 = -e_1$.

The impedance analogous controlled-source circuit model for the push-pull condenser electrostatic-mechanical transducer is obtained by combining the single-ended analogous circuits for the two sides [4]. The combined circuits are shown in Fig. 5(b). The circuit can be implemented in SPICE with three capacitors and four current-controlled current sources as described.

6.4 Mechanoacoustic Transducer [3]

This transducer is a plane piston that radiates an acoustic pressure when it is driven by a mechanical force. Fig. 6(a) illustrates such a transducer, where S is the piston area, f the force driving the piston, u the velocity with which it moves, U the volume velocity emitted from the front of the piston, and $p = p_1 - p_2$ is the pressure difference between the two sides of the piston. The basic equations relating these variables are

$$f = Sp \tag{20}$$

$$U = Su \tag{21}$$

The impedance analogous-circuit model for the mechanoacoustic transducer [4] is given in Fig. 6(b). The circuit can be implemented in SPICE with a voltage-controlled voltage source and a current-controlled current source as described.

7 SPICE MODELING OF ELECTROACOUSTIC MUTUAL COUPLING

Mutual coupling effects between pistons in a common baffle can be difficult to model with electroacoustic

analogous circuits. A rigorous solution to the mutual coupling problem is given in [5], but no electroacoustic models are developed. Approximations are made in [6] and [7] in modeling the coupling effects in vented-box loudspeaker systems, and low-frequency electroacoustic models are given. The controlled-source model presented here predicts the same acoustic mutual coupling mass as the model given in [6].

Fig. 7(a) illustrates two circular pistons in an infinite baffle which are spaced a distance d between centers. Piston 1 has radius a_1 and emits a volume velocity U_1 . Piston 2 has radius a_2 and emits a volume velocity U_2 . The analogous circuit for each piston is obtained by adding a current-controlled current source in parallel with the analogous circuit for the air-load impedance on a piston in an infinite baffle given in [3]. The circuits are shown in Fig. 7(b). The circuits model pressures p_1 and p_2 at the front side of the two pistons as being given by [4]

$$p_1 = (U_1 + k_1 U_2) Z_{A1} \tag{22}$$

$$p_2 = (U_2 + k_2 U_1) Z_{A2} \tag{23}$$

where $k_1 = 3\pi a_1/16d$, $k_2 = 3\pi a_2/16d$, Z_{A1} is the air-load impedance on the front side of piston 1 with piston 2 blocked, Z_{A2} is the air-load impedance on the front side of piston 2 with piston 1 blocked, and U'_1 and U'_2 are the controlling volume velocities labeled on the figure. Equations for the element values in the analogous circuits Z_{A1} and Z_{A2} are given in [3] and are used in the following examples. For a noncircular piston it is common to replace the piston radius a by $\sqrt{S/\pi}$, where S is the area.






Fig. 6. (a) Mechanoacoustic transducer. (b) Controlled-source model of transducer.



(a)



(b)

Fig. 5. (a) Push-pull electrostatic mechanical transducer. (b) Controlled-source model of transducer.

order Butterworth example system design having a lower half-power cutoff frequency of 40 Hz that is described in [6]. A 12-in (0.5 m) driver is assumed for which $a_w = 0.12$ m and $R_E = 6.5 \Omega$. In addition to the parameters given in [6], it will be assumed that $L_E = 1$ mH, $a_p = 1.5$ in (38 mm), $d = 1.5a_w$, and the enclosure quality factor $Q_L = 7$. The remaining parameters for the SPICE simulation are as follows: $Bl = 16.5$ T·m, $M_{MD} = 0.064$ kg, $R_{MS} = 3.72$ N·s/m, $C_{MS} = 2.14 \times 10^{-4}$ m/N, $S_w = 4.5 \times 10^{-2}$ m², $M_{AB} = 2.03$ kg/m⁴, $C_{AB} = 4.03 \times 10^{-7}$ m⁵/N, $R_{AL} = 6.91 \times 10^4$ N·s/m⁵, $M_{AP} = 30.9$ kg/m⁴, $M_{A1P} = 8.37$ kg/m⁴, $R_{A1P} = 3.94 \times 10^4$ N·s/m⁵, $R_{A2P} = 8.93 \times 10^4$ N·s/m⁵, $C_{A1P} = 2.34 \times 10^{-9}$ m⁵/N, $M_{A1W} = 2.66$ kg/m⁴, $R_{A1W} = 3.97 \times 10^3$ N·s/m⁵, $R_{A2W} = 9.00 \times 10^3$ N·s/m⁵, $C_{A1W} = 7.31 \times 10^{-8}$ m⁵/N, $k_w = 0.393$, and $k_p = 0.124$.

The following code has been written to analyze the vented-box system using the PSPICE Circuit Simulator Program. The first line is a title line, which cannot be omitted. The .PROBE command activates the graphics data output subroutine of PSPICE. Methods for obtaining numerical data output from the program are

described in [1] and [2].

VENTED-BOX SYSTEM SIMULATION

```

VEG 1 0 AC 1V
REW 1 2 6.5
LEW 2 3 1M
HBLUW 3 4 VD2 16.5
VIW 4 0 AC 0V
HBLIW 5 0 VD1 16.5
LMMDW 5 6 0.064
RMSW 6 7 3.72
CMSW 7 8 2.14E-4
ESDPW 8 9 17 10 4.5E-2
V2W 9 0 AC 0V
FSDUW 10 17 VD2 4.5E-2
LMABW 10 11 2.03
CABW 11 12 4.03E-7
V3W 12 0 AC 0V
RALW 11 0 6.91E4
LMAW 11 13 30.9
FKPUW 15 13 VD6 0.124
LMA1P 13 15 8.37
RA1P 13 14 3.94E4
CA1P 13 14 2.34E-9
RA2P 14 16 8.93E4
V4W 15 16 AC 0V
V5W 16 0 AC 0V
FKWUP 19 17 VD4 0.393
LMA1W 17 19 2.66
RA1W 17 18 3.97E3
CA1W 17 18 7.31E-8
RA2W 18 20 9E3
V6W 19 20 AC 0V
V7W 20 0 AC 0V
.AC DEC 20 10 10K
.PROBE
.END
    
```




Fig. 8. Vented-box woofer and closed-box midrange.




Fig. 9. Controlled-source analogous circuit for vented-box system.

8 VENTED-BOX LOUDSPEAKER SYSTEM EXAMPLE


An example application of the models to the analysis of vented-box loudspeaker systems is given in this section. Fundamentals of the electroacoustic modeling of vented-box systems are given in [6]. Fig. 8 illustrates the geometry of the system, where it is assumed that the system radiates into half-space. In addition to the woofer, the figure shows a midrange in a closed-box enclosure which is used in a following example. In the figure, a_w is the piston radius of the woofer diaphragm, a_p the piston radius of the port, d the spacing between the centers of diaphragm and port, L the length of the port, and V_{AB} the effective internal box volume. Fig. 9 gives the complete controlled-source analogous circuit model [4] of the woofer and its enclosure with the SPICE node numbers labeled. "Dummy" voltage sources (labeled VNW, where N is an integer) are included in the circuit where needed as ammeters for the current-controlled sources. These sources are assigned a value of 0V in the SPICE analysis code. At a distance r , the magnitude of the low-frequency far-field on-axis acoustic pressure of the system is given by [4]

$$|p(r)| = \frac{\rho_0 f}{r} |U_B| \tag{24}$$

where U_B is the total system output volume velocity, $\rho_0 = 1.18 \text{ kg/m}^3$ is the air density, and f is the frequency.

The electroacoustic variables in the circuit of Fig. 9 are as follows:

- e_g = voice-coil voltage
- i_c = voice-coil current
- u_w = diaphragm velocity
- U_w = diaphragm volume velocity
- U_p = port volume velocity
- U_L = enclosure air-leak volume velocity
- U_B = total system output volume velocity, $= U_D + U_P + U_L$
- p_F = acoustic pressure at front of diaphragm
- p_B = acoustic pressure at back of diaphragm.



The circuit element parameters are as follows:

- R_E = voice-coil resistance
- L_E = voice-coil inductance
- Bl = Bl product of voice-coil and magnet assembly
- M_{MD} = mechanical mass of diaphragm
- R_{MS} = damping resistance of diaphragm suspension
- C_{MS} = compliance of diaphragm suspension
- S_w = piston area of woofer diaphragm, $= \pi a_w^2$
- S_p = piston area of port, $= \pi a_p^2$
- M_{AB} = air-load mass on back of woofer diaphragm [8], $= \rho_0 d S_w / 3 S_b^2 + 8 \rho_0 (1 - S_w / S_b) / 3 \pi \sqrt{\pi S_w}$, where d is box depth and S_b is inside area of panel on which driver is mounted
- C_{AB} = acoustic compliance of enclosure volume V_{AB} , $= V_{AB} / \rho_0 c^2$
- R_{AL} = acoustic resistance modeling air leaks in enclosure
- M'_{AP} = $M_{AP} - M_{A1P}$, where $M_{AP} = 1 / \omega_B^2 C_{AB}$, ω_B is radian Helmholtz tuning frequency, and M_{A1P} is external port air-load mass given below
- k_w = mutual coupling coefficient from port to diaphragm, $= 3 \pi a_w / 16 d$
- k_p = mutual coupling coefficient from diaphragm to port, $= 3 \pi a_p / 16 d$

M_{A1W} , R_{A1W} , R_{A2W} , and C_{A1W} are elements that model air-load impedance on the diaphragm, and M_{A1P} , R_{A1P} , R_{A2P} , and C_{A1P} are elements that model air-load impedance on the port. The air-load impedance element values are [3]

$$\begin{aligned} M_{A1} &= 8 \rho_0 / 3 \pi^2 a \\ R_{A2} &= \rho_0 c / \pi a^2 \\ R_{A1} &= 128 \rho_0 c / 9 \pi^3 a^2 - R_{A2} \\ C_{A1} &= 5.94 a^3 \rho_0 c^2 \end{aligned}$$

where $a = a_w$ for the woofer diaphragm elements and $a = a_p$ for the port elements. The flanged end correction for the acoustic mass of the port is not a part of M'_{AP} because it is modeled by M_{A1P} .

The following numerical example is based on a fourth-




Fig. 7. (a) Mutual coupling between pistons in a common baffle. (b) Controlled-source analogous circuit for air loads on pistons, including effects of mutual coupling.

Fig. 10 illustrates the normalized on-axis sound-pressure-level (SPL) responses of the system as displayed by the PROBE graphics subroutine of PSPICE. The figure shows the overall SPL response of the system, the SPL response of the diaphragm only, and the SPL response of the port only. These responses are calculated from Eq. (22) with $r = 1$ m from the equation $SPL = 20 \log(|p|/p_{ref})$, where $p_{ref} = 2 \times 10^{-5}$ Pa. The familiar Butterworth rolloff at the low end in the response is predicted accurately. The rolloff at the high end must be considered to be only approximate because the analogous circuits do not model resonances in the driver diaphragm which can cause erratic response near the upper cutoff frequency. The analysis also predicts the phase responses of the system, which are omitted here.

9 CONDENSER MICROPHONE EXAMPLE

An example application of the models to the analysis of condenser microphones is given in this section. Fundamentals of the condenser microphone are discussed in [3], [9], and [10]. Fig. 11 illustrates the geometry of the microphone. The assumed specifications are aluminum diaphragm material of density 2700 kg/m^3 , diaphragm, electrode, and front air cavity radii $a = 0.01$ m, diaphragm thickness $t = 4 \times 10^{-5}$ m, diaphragm to back plate spacing $x_0 = 4 \times 10^{-5}$ m, diaphragm tension $T = 2 \times 10^4$ N/m, back air cavity volume equal to 100 times front air cavity volume, polarizing voltage $E = 300$ V, and a total quality factor of 1.0 at the fundamental resonance. The mechanical mass M_{MD} and compliance C_{MD} [10] of the diaphragm are $M_{MD} = \frac{1}{3}\pi a^2 t \rho_1 = 4.52 \times 10^{-5}$ kg and $C_{MD} = 1/8\pi T = 1.99 \times 10^{-6}$ m/N. The acoustic compliances of the front and back cavities are $C_{AB1} = V_{AF}/\rho_0 c^2 = 8.95 \times 10^{-14} \text{ N}\cdot\text{m}^3$ and $C_{AB2} = 100 \times C_{AB1} = 8.95$

$\times 10^{-12} \text{ N}\cdot\text{m}^3$. A diaphragm mechanical damping resistance $R_{MD} = 0.178 \text{ N}\cdot\text{s/m}$ is assumed. For proper damping, the screen perforations in the back plate must have an acoustic resistance $R_{AS} = 5.22 \times 10^7 \text{ N}\cdot\text{s/m}^5$. The acoustic mass of the screen perforations is taken to be $M_{AS} = 132 \text{ kg/m}^4$. The electrical capacitance of the microphone is $C_{E0} = \epsilon_0 \pi a^2/x_0 = 69.5 \text{ pF}$. A load resistance of $R_L = 20 \text{ M}\Omega$ is used for the simulation.

Fig. 12 shows the complete controlled source model [4] of the microphone. In the acoustical circuit, the pressure generator labeled p_B models the acoustic pressure in an incident plane wave in the absence of the microphone. The volume-velocity generator in the acoustical circuit labeled p_B/R_{A2} models the effects of reflections from the diaphragm [11] for a plane wave that is incident normally on the diaphragm. When the direction of incidence is in the plane of the diaphragm, this generator is omitted. The resistor labeled R_{AL} in the acoustical circuit is included to prevent nodes 4, 5, and 6 from being floating nodes at direct current. Without this resistor, SPICE would not run. The value of R_{AL} for the following simulations has been chosen large enough so that it is an open circuit for all practical purposes. The following SPICE code has been written to calculate the output voltage from the example con-




Fig. 11. Geometry for condenser microphone example.




Fig. 10. Calculated on-axis SPLs at 1 m for vented-box woofer system. (a)—total system; (b)—woofer alone; (c)—port alone.

denser microphone as a function of frequency for an incident pressure amplitude of $0.1 \mu\text{bar}$.

CONDENSER MICROPHONE EXAMPLE

```

VPB 1 0 AC 0.1V
LMA1 1 3 23
RA1 1 2 6.54E5
CA1 1 2 3.87E-11
RA2 2 3 1.3E6
GPB 0 3 1 0 7.72E-7
FSDUD 4 3 V1 3.14E-4
CAB1 4 0 8.95E-14
LMAS 4 5 132
RAS 5 6 5.22E7
CAB2 6 0 8.95E-12
RAL 6 0 1E12
FECMI 7 0 V2 14.9
CMD 7 0 1.99E-6
V1 7 8 AC 0V
LMMD 8 9 4.52E-5
RMD 9 10 0.178
ESDPD 10 0 3 4 3.14E-4
RL 11 0 20E6
V2 11 12 AC 0V
CEO 12 0 69.5E-12
FECEO 0 12 V1 5.22E-4
.AC DEC 50 100 100K
.PROBE
.END
    
```

Fig. 13 shows the calculated output voltages in decibels from the microphone as a function of frequency for the incident pressure of $0.1 \mu\text{bar}$. The voltage output is shown for two cases—with the volume-velocity generator p_B/R_{A2} that models reflections from the diaphragm omitted (lower curve) and with the generator

in the circuit (upper curve). The curves model the range of expected responses from the microphone for angles of incidence between 0° and 90° .

10 LOUDSPEAKER CROSSOVER NETWORK EXAMPLE

This example illustrates the application of SPICE modeling to the evaluation of crossover networks for a woofer and a midrange. The vented-box loudspeaker modeled in Sec. 8 is used as the woofer in the system. For the midrange, a closed-box driver is assumed, as illustrated in Fig. 8. Details of closed-box loudspeaker design are described in [12]. To simplify the example, the mutual coupling between midrange and woofer and between midrange and port are neglected. In addition, it is assumed that the midrange has the same efficiency as the woofer and has the following parameters: diaphragm piston radius $a_m = 0.04 \text{ m}$, fundamental closed-box resonance frequency $f_c = 300 \text{ Hz}$, closed-box electrical quality factor $Q_{EC} = 1.0$, closed-box mechanical quality factor $Q_{MC} = 4$, closed-box total quality factor $Q_{TC} = 0.8$, system total volume compliance $V_{AT} = 0.323 \text{ liters}$, system compliance ratio $\alpha = 3$, driver mechanical quality factor $Q_{MS} = 3$, and diaphragm rear mass loading factor $B = 0.449$. Fig. 14 gives the complete controlled-source analogous circuit model [4] of the midrange and its enclosure with the SPICE node numbers labeled. The figure shows a second-order high-pass crossover network preceding the midrange voice coil. A second-order low-pass crossover network is shown for the woofer. This network connects to the woofer system model circuit of Fig. 9.

Fig. 14 shows two volume-velocity summing networks, one used to add the volume-velocity outputs




Fig. 12. Controlled-source analogous circuit for condenser microphone.

from the woofer system and the midrange and the other to subtract the volume-velocity outputs. These networks allow the simultaneous simulation of the system responses with the voice coils of the two drivers connected in electrical phase and out of electrical phase. If mutual coupling effects between the drivers are included, the simultaneous simulations would not be possible, that is, the voice-coil connections of one driver would have to be reversed at the crossover network for the out-of-phase simulation.

The circuit element values in Fig. 14 are as follows: voice-coil resistance $R_E = 6.5 \Omega$, voice-coil inductance

$L_E = 0.2 \text{ mH}$, front air-load acoustic mass $M_{A1M} = 8\rho_0/3\pi^2 a_m = 7.97 \text{ kg/m}^4$, front air-load acoustic resistance $R_{A2M} = \rho_0 c/\pi a_m^2 = 8.10 \times 10^4 \text{ N}\cdot\text{s/m}^5$, front air-load acoustic resistance $R_{A1M} = (128/9\pi^2 - 1) R_{A2M} = 3.57 \times 10^4 \text{ N}\cdot\text{s/m}^5$, front air-load acoustic compliance $C_{A1M} = 5.94 a_m^3/\rho_0 c^2 = 2.71 \times 10^{-9} \text{ N}\cdot\text{m}^3$, enclosure acoustic compliance $C_{AB} = (1 + \alpha)V_{AT}/\alpha\rho_0 c^2 = 3.07 \times 10^{-9} \text{ N}\cdot\text{m}^3$, rear air-load acoustic mass $M_{AB} = B\rho_0/\pi a_m = 4.22 \text{ kg/m}^4$, diaphragm and voice-coil mechanical mass $M_{MD} = [\rho_0 c^2/(2\pi f_C)^2 V_{AT} - M_{A1M} - M_{AB}]\pi^2 a_m^4 = 2.78 \times 10^{-3} \text{ kg}$, driver mechanical resistance $R_{MS} = \pi a_m^4 \rho_0 c^2/s f_C Q_{MS} V_{AT} \sqrt{1 + \alpha} =$




Fig. 13. Calculated output voltages from condenser microphone. (a)— 0° angle of incidence; (b)— 90° angle of incidence.




Fig. 14. Controlled-source analogous circuit for midrange and crossover networks.

0.971 N·s/m, box acoustical resistance $R_{AB} = R_{MS} (\sqrt{1 + \alpha Q_{MS}/Q_{MC}} - 1) / \pi^2 a_m^4 = 1.92 \times 10^4 \text{ N·s/m}^5$, driver mechanical compliance $C_{MS} = (1 + \alpha) V_{AT} / \pi^2 a_m^4 \rho_0 c^2 = 3.64 \times 10^{-4} \text{ m/N}$, diaphragm piston area $S_D = \pi a_m^2 = 5.03 \times 10^{-3} \text{ m}^2$, and driver Bl product is $Bl = \pi a_m^2 [R_E \pi_0 c^2 / (2\pi f_C V_{AT} Q_{EC})]^{1/2} = 6.15 \text{ T·m}$.

The crossover frequency has been chosen to be $f_{c0} = 450 \text{ Hz}$. This is approximately the frequency at which the wavelength is equal to the circumference of the woofer diaphragm. The crossover network elements have been calculated for a quality factor $Q_{co} = 0.5$ under the assumption that each network has a resistive load equal to the voice-coil resistance R_E of the driver. The element values are given by $C_1 = C_2 = Q_{co} / 2\pi f_{c0} R_E = 27.2 \text{ } \mu\text{F}$ and L_1 and $L_2 = R_E / 2\pi f_{c0} Q_{co} = 4.60 \text{ mH}$. Resistors R_1 and R_2 in the circuit represent the dc resistances of L_1 and L_2 , respectively. To calculate these, it was assumed that the resistance of the inductors varies as the square root of the inductance and that a 2.5-mH inductor has a dc resistance of 1 Ω . This is a typical value for an air-core inductor wound with #18 wire. The following SPICE code has been written to calculate the on-axis SPL of the system, where the omitted woofer code is the same as that given in Sec. 8.

CROSSOVER NETWORK EXAMPLE

```
VEG 21 0 AC 1V
L1 21 22 4.6M
R1 22 1 1.36
C1 1 0 27.2U
C2 21 23 27.2U
L2 23 24 4.6M
R2 24 0 1.36
REM 23 25 6.5
LEM 25 26 0.37M
```

```
HBLUM 26 27 V2M 6.15
V1M 27 0 AC 0V
HBLIM 28 0 V1M 6.15
LMMDM 28 29 2.82M
RMSM 29 30 1.46
CMSM 30 31 364U
ESMPM 31 32 36 33 5.03M
V2M 32 0 AC 0V
FSMUM 33 36 V2M 5.03M
LMABM 33 34 2.89
CABM 34 35 3.07N
RAL 34 35 1E12
RABM 35 0 57.6K
RA1M 36 37 35.7K
RA2M 37 38 81K
CA1M 36 37 2.71N
LMA1M 36 38 7.97
V3M 38 0 AC 0V
.
. woofer code
.
FW1 0 39 V3W 1
FM1 0 39 V3M 1
VSUM 39 0 AC 0V
FW2 0 40 V3M 1
FM2 40 0 V3W 1
VDIF 40 0 AC 0V
.AC DEC 50 10 10K
.PROBE
.END
```

The on-axis SPL response of the system can be calculated from Eq. (24) with U_B replaced by $U_{sum} = U_B + U_M$, where U_M is the volume velocity emitted by the midrange diaphragm. Fig. 15 shows the cal-




Fig. 15. Calculated on-axis SPLs at 1 m for woofer and midrange. (a)—pressure sum response; (b)—pressure difference response; (c)—woofer system alone with crossover network; (d)—midrange alone with crossover network.

culated responses for $r = 1$ m for the woofer plus crossover network, the midrange plus crossover network, the total system output with the drivers operating in phase, and the total system output with the drivers operating out of phase. The out-of-phase connection results in the flattest overall frequency response. However, some of the typical problems that are normally encountered in crossover system design are evident in the plot. The calculated input impedance of the system is shown in Fig. 16.

11 CONCLUSIONS

The SPICE electrical circuit simulator software is a powerful tool for the computer-aided design of electroacoustic systems, including microphones, loudspeakers, and crossover networks. With SPICE, complex systems containing multiple electrical, mechanical, and acoustical analogous circuits can be analyzed quickly and efficiently. The four controlled sources of SPICE make it possible to model electro-mechano-acoustic transducers using either impedance analogous circuits, mobility analogous circuits, or a combination of both. This eliminates restrictions imposed on the analogous circuits when the usual transformer is used in modeling transducers. The scope of this engineering report has been limited to an abbreviated overview of the applications of SPICE with three illustrative examples. Only impedance analogous circuits of the transducers have been used. These models can be converted easily to mobility models.

12 REFERENCES

[1] W. Banzhaf, *Computer-Aided Circuit Analysis Using SPICE* (Prentice-Hall, Englewood Cliffs, NJ,

1989).

[2] P. W. Tuinenga, *SPICE, A Guide to Circuit Simulation and Analysis Using PSpice* (Prentice-Hall, Englewood Cliffs, NJ, 1988).

[3] L. L. Beranek, *Acoustics* (McGraw-Hill, New York, 1954).

[4] W. M. Leach, Jr., *Audio Engineering Class Notes* (Georgia Tech Photographic and Printing Center, Atlanta, 1991).

[5] O. Jacobsen, "Some Aspects of the Self and Mutual Radiation Impedance Concept with Respect to Loudspeakers," *J. Audio Eng. Soc.*, vol. 24, pp. 82-92 (1976 Mar.).

[6] R. H. Small, "Vented-Box Loudspeaker Systems, Parts I-IV," *J. Audio Eng. Soc.*, vol. 21, pp. 363-372 (1973 June); pp. 438-444 (1973 July/Aug.); pp. 549-554 (1973 Sept.); pp. 635-639 (1973 Oct.).

[7] B. N. Locanthi, "Application of Electric Circuit Analogies to Loudspeaker Design Problems," *J. Audio Eng. Soc.*, vol. 19, pp. 778-785 (1971 Oct.).

[8] W. M. Leach, Jr., "Electroacoustic-Analogous Circuit Models for Filled Enclosures," *J. Audio Eng. Soc.*, vol. 37, pp. 586-592 (1989 July/Aug.).

[9] F. V. Hunt, *Electroacoustics* (Am. Inst. of Physics for the Acoust. Soc. Am., 1982).

[10] P. M. Morse, *Vibration and Sound* (Am. Inst. of Physics for Acoust. Soc. Am., 1976).

[11] W. M. Leach, Jr., "On the Electroacoustic-Analogous Circuit for a Plane Wave Incident on the Diaphragm of a Free-Field Pressure Microphone," *J. Audio Eng. Soc. (Engineering Reports)*, vol. 38, pp. 566-568 (1990 July/Aug.).

[12] R. H. Small, "Closed-Box Loudspeaker Systems, Parts I and II," *J. Audio Eng. Soc.*, vol. 20, pp. 798-808 (1972 Dec.); vol. 21, pp. 11-18 (1973 Jan./Feb.).





Fig. 16. Calculated input impedance for woofer and midrange system.

THE AUTHOR

W. Marshall Leach, Jr., received B.S. and M.S. degrees in electrical engineering from the University of South Carolina, Columbia, in 1962 and 1964, and a Ph.D. degree in electrical engineering from the Georgia Institute of Technology in 1972.

In 1964 he worked at the National Aeronautics and Space Administration in Hampton, VA. From 1965 to

1968 he served as an officer in the U.S. Air Force. Since 1972 he has been a faculty member at the Georgia Institute of Technology, where he is presently professor of electrical engineering. Dr. Leach teaches courses in applied electromagnetics and electronic design. He is a fellow of the Audio Engineering Society and a senior member of the IEEE.